

§ 3 Ajustements au sens des moindres carrés

§ 3.1 Ajustement par une combinaison linéaire de fonctions à une variable

■ Exemple: Ajustement d'une droite


■ Énoncé du problème d'ajustement au sens des moindres carrés

Une expérience de physique consiste à mesurer le volume y d'un gaz, à pression constante, en fonction de la température x :

Get::noopen : Cannot open Tableaux`. >>

```
afficheTableau[{x [°C], y [cm³]}, None, {{x₁, x₂, ..., xₙ}, {y₁, y₂, ..., yₙ}}]
```


n désigne le nombre de mesures effectuées. Figure pour $n = 5$:



Pour des raisons physiques, on postule que y est une fonction affine de x :

$$y = ax + b$$

Étant donné l'imprécision des mesures, la loi théorique n'est vérifiée qu'approximativement:



Les écarts suivants sont appelés "résidus"

$$e_1 = y_1 - (a x_1 + b), \quad e_2 = y_2 - (a x_2 + b), \quad \dots, \quad e_n = y_n - (a x_n + b)$$

"Ajuster la droite au sens des moindres carrés" signifie qu'il faut déterminer a, b de manière que la somme des carrés des résidus soit minimale

$$e_1^2 + e_2^2 + \dots + e_n^2 = \min$$

■ Réduction du problème d'ajustement à un problème de projection

$$e_1^2 + e_2^2 + \dots + e_n^2 = (y_1 - a x_1 - b)^2 + (y_2 - a x_2 - b)^2 + \dots + (y_n - a x_n - b)^2 =$$

$$\left\| \begin{pmatrix} y_1 - a x_1 - b \\ y_2 - a x_2 - b \\ \vdots \\ y_n - a x_n - b \end{pmatrix} \right\|^2 = \left\| \begin{pmatrix} y_1 \\ y_2 \\ \vdots \\ y_n \end{pmatrix} - a \begin{pmatrix} x_1 \\ x_2 \\ \vdots \\ x_n \end{pmatrix} - b \begin{pmatrix} 1 \\ 1 \\ \vdots \\ 1 \end{pmatrix} \right\|^2$$

Autrement dit, dans \mathbb{R}^n , on donne le vecteur

$$\vec{y} = \begin{pmatrix} y_1 \\ y_2 \\ \vdots \\ y_n \end{pmatrix}$$

et le sous-espace G , engendré par les vecteurs

$$\vec{g}_1 = \begin{pmatrix} x_1 \\ x_2 \\ \vdots \\ x_n \end{pmatrix}, \quad \vec{g}_2 = \begin{pmatrix} 1 \\ 1 \\ \vdots \\ 1 \end{pmatrix}, \quad \text{dont les éléments sont} \quad \vec{g} = a \vec{g}_1 + b \vec{g}_2$$

Déterminez les nombres a, b tels que

$$\left(\left\| \vec{y} - \vec{g} \right\| \right)^2 = \min$$

Nous avons étudié, dans le § 2, que la solution est la projection orthogonale \vec{g} de \vec{y} sur G déterminée par les conditions suivantes

$$\begin{aligned} \vec{g} &= a \vec{g}_1 + b \vec{g}_2 \\ \vec{e} &= \vec{y} - \vec{g} = \vec{y} - a \vec{g}_1 - b \vec{g}_2 \\ \vec{e} \cdot \vec{g}_1 &= 0 \quad \text{et} \quad \vec{e} \cdot \vec{g}_2 = 0 \end{aligned}$$

On obtient ainsi les équations normales

$$\begin{aligned} (\vec{y} - a \vec{g}_1 - b \vec{g}_2) \cdot \vec{g}_1 &= 0 \quad \text{et} \quad (\vec{y} - a \vec{g}_1 - b \vec{g}_2) \cdot \vec{g}_2 = 0 \\ \vec{y} \cdot \vec{g}_1 - a \vec{g}_1 \cdot \vec{g}_1 - b \vec{g}_2 \cdot \vec{g}_1 &= 0 \quad \text{et} \quad \vec{y} \cdot \vec{g}_2 - a \vec{g}_1 \cdot \vec{g}_2 - b \vec{g}_2 \cdot \vec{g}_2 = 0 \\ \vec{g}_1 \cdot \vec{g}_1 a + \vec{g}_2 \cdot \vec{g}_1 b &= \vec{y} \cdot \vec{g}_1 \quad \text{et} \quad \vec{g}_1 \cdot \vec{g}_2 a + \vec{g}_2 \cdot \vec{g}_2 b = \vec{y} \cdot \vec{g}_2 \\ \begin{pmatrix} \vec{g}_1 \cdot \vec{g}_1 & \vec{g}_2 \cdot \vec{g}_1 \\ \vec{g}_1 \cdot \vec{g}_2 & \vec{g}_2 \cdot \vec{g}_2 \end{pmatrix} \begin{pmatrix} a \\ b \end{pmatrix} &= \begin{pmatrix} \vec{y} \cdot \vec{g}_1 \\ \vec{y} \cdot \vec{g}_2 \end{pmatrix} \end{aligned}$$

Les coefficients du système sont

$$\begin{aligned} \vec{g}_1 \cdot \vec{g}_1 &= x_1^2 + x_2^2 + \dots + x_n^2 = \sum_{i=1}^n x_i^2 \\ \vec{g}_2 \cdot \vec{g}_1 &= x_1 + x_2 + \dots + x_n = \sum_{i=1}^n x_i = \vec{g}_1 \cdot \vec{g}_2 \\ \vec{g}_2 \cdot \vec{g}_2 &= 1^2 + 1^2 + \dots + 1^2 = n \\ \vec{y} \cdot \vec{g}_1 &= Y_1 x_1 + Y_2 x_2 + \dots + Y_n x_n = \sum_{i=1}^n x_i Y_i \\ \vec{y} \cdot \vec{g}_2 &= Y_1 + Y_2 + \dots + Y_n = \sum_{i=1}^n Y_i \end{aligned}$$

Les équations normales prennent la forme

$$\begin{pmatrix} \sum_{i=1}^n x_i^2 & \sum_{i=1}^n x_i \\ \sum_{i=1}^n x_i & n \end{pmatrix} \begin{pmatrix} a \\ b \end{pmatrix} = \begin{pmatrix} \sum_{i=1}^n x_i Y_i \\ \sum_{i=1}^n Y_i \end{pmatrix}$$

■ Exemple numérique et résolution des équations normales

```
x = {1, 3, 4, 6, 8, 9, 11, 14};  
y = {1, 2, 4, 4, 5, 7, 8, 9};
```

```
Get::noopen : Cannot open Tableaux`. >>
```

```
afficheTableau[{x, y}, None, {{1, 3, 4, 6, 8, 9, 11, 14}, {1, 2, 4, 4, 5, 7, 8, 9}}]
```

```
n = Length[x]
```

```
8
```

```
 $\vec{g}_1 = \vec{x}$ 
```

```
g2 = Table[1, {n}]
```

```
{1, 1, 1, 1, 1, 1, 1, 1}
```

```
m = {{x.x, x.g2}, {g2.x, g2.g2}}; MatrixForm[m]
```

```
 $\begin{pmatrix} 524 & 56 \\ 56 & 8 \end{pmatrix}$ 
```


```
v = {y.x, y.g2}
```

```
{364, 40}
```

```
{a, b} = LinearSolve[m, v]
```

```
{ 7/11, 6/11 }
```

Les valeurs de a et b peuvent aussi être estimées approximativement à partir d'un graphique:



Finalement, nous avons résolu le problème en 5 étapes:

1. formulation du problème d'ajustement;
2. réduction du problème d'ajustement à un problème de moindres carrés;
3. réduction du problème de moindres carrés à un problème de projection orthogonale;
4. réduction du problème de projection orthogonale à un système d'équations normales;
5. résolution du système d'équations.

■ Autre méthode de calcul : la fonction `Fit[...]` de *Mathematica*

?Fit

`Fit[data, funs, vars]` trouve un ajustement aux moindres carrés d'une liste de données comme combinaison linéaire des fonctions `funs` des variables `vars`. Le format des données peut être `{{x1, y1, ..., f1}, {x2, y2, ..., f2}, ...}`, où le nombre de coordonnées `x, y, ...` est égal au nombre de variables de la liste `vars`. Le format des données peut aussi être `{f1, f2, ...}`, où l'on considère qu'une seule coordonnée prend les valeurs 1, 2, L'argument `funs` peut être une liste quelconque de fonctions qui ne dépendent que des objets `vars`.

```
donnees = Transpose[{x, y}]
```

```
{{1, 1}, {3, 2}, {4, 4}, {6, 4}, {8, 5}, {9, 7}, {11, 8}, {14, 9}}
```

Comme la lettre x est déjà utilisée pour désigner la liste des abscisses, on doit choisir une autre lettre pour désigner la variable. Si on choisit comme variable le symbole t , les fonctions de base sont

$$f_1(t) = t, \quad f_2(t) = 1$$

et la fonction que l'on cherche est une combinaison linéaire des deux fonctions de base

$$a t + b = a f_1(t) + b f_2(t)$$

```
f[t_] = Fit[donnees, {t, 1}, t]
```

$$0.545455 + 0.636364 t$$

En prenant un autre exemple, **Fit[donnees, {t², 1/t}, t]** signifie que l'on combine les fonctions $t \mapsto t^2$ et $t \mapsto \frac{1}{t}$ pour obtenir la fonction d'ajustement $t \mapsto a \cdot t^2 + b \cdot \frac{1}{t}$ dans laquelle a et b sont les deux inconnues du problème d'ajustement.

Remarque: pour définir la fonction, utiliser une assignation immédiate "=" plutôt qu'une assignation différée ":=".

■ Exercice 3.1 - 1 [Sans ordinateur] Propriétés des résidus

On ajuste la droite $y = a x + b$ aux n points de mesure $(x_1, y_1), \dots, (x_n, y_n)$ au sens des moindres carrés. On note $e_i = y_i - a x_i - b$ les résidus pour $i \in \{1, 2, \dots, n\}$. Démontrez que

- a) la somme des résidus est nulle:

$$\sum_{i=1}^n e_i = 0$$

- b) la somme des résidus, pondérée par les abscisses, est nulle:

$$\sum_{i=1}^n x_i e_i = 0$$

Indication: Il s'agit de conséquences immédiates des équations normales.

■ Exercice 3.1 - 2 [Sans ordinateur] Ajustements à un paramètre

Reprenons les données de l'exemple numérique:

x	1	3	4	6	8	9	11	14
y	1	2	4	4	5	7	8	9

- a) Ajustez la droite $y = a x$ à ces données, au sens des moindres carrés.

Directive: répétez toute la démarche en 5 étapes.

- b) Ajustez la droite $y = b$ à ces données, au sens des moindres carrés.

Interprétez le résultat obtenu.

■ Exercice 3.1- 3 Ajustement d'une parabole


Voici une statistique des taux de natalité aux USA. Le taux de natalité est le nombre d'enfants nés vivants en une année pour mille habitants

```
x = Range[1915, 1955, 5];
```

```
y = {25.0, 23.7, 21.3, 18.9, 16.9, 17.9, 19.5, 23.6, 24.6};
```

x [an]	1915	1920	1925	1930	1935	1940	1945	1950	1955
y [pour mille par an]	25.	23.7	21.3	18.9	16.9	17.9	19.5	23.6	24.6


```
ListPlot[Transpose[{x, y}], Joined -> True, AxesOrigin -> {1910, 15}]
```



On y remarque l'effet de la crise de 1929. Fixons l'origine du temps au premier juillet 1935

x [an]	-20	-15	-10	-5	0	5	10	15	20
y [pour mille par an]	25.	23.7	21.3	18.9	16.9	17.9	19.5	23.6	24.6

```
ListPlot[Transpose[{x, y}], Joined → True, AxesOrigin → {0, 15}]
```



On demande d'ajuster la parabole $y = a x^2 + b x + c$ à ces données.

Directive: répétez toute la démarche en 5 étapes.

[Avec *Mathematica*] Faites une représentation graphique.

■ Exercice 3.1 - 4 Droite de régression de x en fonction de y

Parmi les étudiants de sexe masculin d'une université, on en a tiré 12 au hasard. On a mesuré leurs masses x en kilogrammes et leurs tailles respectives y en centimètres:

```
x = {70, 68, 63, 72, 60, 66, 70, 74, 65, 62, 67, 65};
y = {155, 152, 150, 180, 135, 156, 168, 178, 160, 132, 145, 139};
```

x [kg]	70	68	63	72	60	66	70	74	65	62	67	65
y [cm]	155	152	150	180	135	156	168	178	160	132	145	139

Ajustons d'abord la droite $y = a x + b$ au sens des moindres carrés; cette droite est aussi appelée droite de régression de y en fonction de x ; on obtient

```
Fit[Transpose[{x, y}], {t, 1}, t]
```

```
- 60.7461 + 3.21565 t
```

Ajustons ensuite la droite $x = m y + p$ au sens des moindres carrés; cette droite est aussi appelée droite de régression de x en fonction de y ; on obtient

```
Fit[Transpose[{y, x}], {t, 1}, t]
31.1078 + 0.231733 t
```

Questions:

- Pourquoi les deux fonctions affines ne sont-elles pas réciproques ?
- Ajustez la droite $y = a x + b$ en minimisant la somme des carrés des résidus horizontaux.

§ 3.2 Moindres carrés pondérés [Pour OS PAM seulement]

On peut donner plus d'importance à certaines mesures en les répétant. Reprenons l'exemple numérique du § 3.1 en répétant deux fois la première mesure et trois fois la dernière:

```
x = {1, 1, 3, 4, 6, 8, 9, 11, 14, 14, 14};
y = {1, 1, 2, 4, 4, 5, 7, 8, 9, 9, 9};
```

x	1	1	3	4	6	8	9	11	14	14	14
y	1	1	2	4	4	5	7	8	9	9	9

Alors, dans l'expression des moindres carrés apparaissent des coefficients appelés "poids":

$$2 (1 - a - b)^2 + (2 - a - b)^2 + (4 - a - b)^2 + (4 - a - b)^2 + (5 - a - b)^2 + (7 - a - b)^2 + (8 - a - b)^2 + 3 (9 - a - b)^2$$

Avec de tels poids, la réponse obtenue est différente:

```
donnees = Transpose[{x, y}];
Fit[donnees, {t, 1}, t]
0.579315 + 0.619147 t
```

Plus généralement, on peut définir un problème de "moindres carrés pondérés"

$$c_1 e_1^2 + c_2 e_2^2 + \dots + c_n e_n^2 = c_1 (y_1 - a x_1 - b)^2 + c_2 (y_2 - a x_2 - b)^2 + \dots + c_n (y_n - a x_n - b)^2$$

où les poids c_1, c_2, \dots, c_n sont des nombres réels positifs (pas nécessairement entiers).

La méthode de résolution consiste à redéfinir un nouveau produit scalaire pondéré

$$\left\langle \vec{u} \mid \vec{v} \right\rangle = c_1 u_1 v_1 + c_2 u_2 v_2 + \dots + c_n u_n v_n$$

Avec ce nouveau produit scalaire pondéré, la norme des vecteurs ne sera plus la norme euclidienne usuelle mais sera aussi pondérée par les coefficients souhaités:

$$\left(\parallel \vec{e} \parallel \right)^2 = \left\langle \vec{e} \mid \vec{e} \right\rangle = c_1 e_1^2 + c_2 e_2^2 + \dots + c_n e_n^2$$

Pour que la norme de ce vecteur soit minimale, il suffit de calculer la projection orthogonale en remplaçant le produit scalaire usuel par le nouveau produit scalaire pondéré.

Dans notre exemple numérique,

```
x = {1, 3, 4, 6, 8, 9, 11, 14};
y = {1, 2, 4, 4, 5, 7, 8, 9};
c = {2, 1, 1, 1, 1, 1, 1, 3};
Clear[u, v, ps];
```

```

ps[u_List, v_List] := (c * u) . v

n = Length[x]

8

g2 = Table[1, {n}]

{1, 1, 1, 1, 1, 1, 1, 1}

m = {{ps[x, x], ps[x, g2]}, {ps[g2, x], ps[g2, g2]}}; MatrixForm[m]


$$\begin{pmatrix} 917 & 85 \\ 85 & 11 \end{pmatrix}$$


v = {ps[y, x], ps[y, g2]}

{617, 59}

{a, b} = LinearSolve[m, v]; N[{a, b}]

{0.619147, 0.579315}

```

■ Exercice 3.2-1 [Pour OS PAM seulement]

Une statistique a montré que les couples (x, y) ne prenaient que les valeurs données dans le tableau qui suit, avec les fréquences correspondantes c

```

x = {0, 1, 2, 3, 4, 5, 6, 7};
y = {100, 103, 107, 109, 111, 114, 117, 121};
c = {0.05, 0.1, 0.15, 0.2, 0.2, 0.15, 0.1, 0.05};

```

x	0	1	2	3	4	5	6	7
y	100	103	107	109	111	114	117	121
c	0.05	0.1	0.15	0.2	0.2	0.15	0.1	0.05

Ajustez la droite $y = ax + b$ au sens des moindres carrés pondérés en prenant pour poids les fréquences c_i .

Directive: répétez toute la démarche en 5 étapes.

§ 3.3 Ajustement par une combinaison linéaire de fonctions à plusieurs variables

■ Exemple: Ajustement d'un plan

Le graphique d'une fonction de deux variables, par exemple


$$z = f(x, y) = -2x - 3y + 25$$

se présente comme une surface dans l'espace.

```

Plot3D[-2 x - 3 y + 25, {x, 0, 5}, {y, 0, 5},
ViewPoint -> {3, 1, 1}, AxesLabel -> {"x", "y", "z"}]

```

Dans le cas particulier où la fonction est affine

$$z = f(x, y) = ax + by + c$$

la surface est un plan dont l'équation est

$$ax + by - z + c = 0$$

Etant donné un "nuage de points" et un plan, on appelle "résidus" les écarts entre les points et le plan mesurés parallèlement à l'axe des z ; on cherche le plan tel que la somme des carrés des résidus soit minimale.

Choisissons comme variables les symboles s et t . On peut considérer la fonction cherchée comme étant une combinaison linéaire des trois fonctions de base suivantes

$$f_1(s, t) = s, \quad f_2(s, t) = t, \quad f_3(s, t) = 1$$

En effet,

$$af_1(s, t) + bf_2(s, t) + cf_3(s, t) = as + bt + c$$

■ Exercice 3.3-1

On a choisi 12 garçons au hasard et on a mesuré

x = l'âge, en années;

y = le tour de taille, en cm;

z = le tour de poitrine, en cm.

$\mathbf{x} = \{8, 10, 6, 11, 8, 7, 10, 9, 10, 6, 12, 9\};$

$\mathbf{y} = \{57, 59, 49, 62, 51, 50, 55, 48, 52, 42, 61, 57\};$

$\mathbf{z} = \{64, 71, 53, 67, 55, 58, 77, 57, 56, 51, 76, 68\};$

\mathbf{x} [an]	8	10	6	11	8	7	10	9	10	6	12	9
\mathbf{y} [cm]	57	59	49	62	51	50	55	48	52	42	61	57
\mathbf{z} [cm]	64	71	53	67	55	58	77	57	56	51	76	68

Ajustez la fonction $z = ax + by + c$ aux données, au sens des moindres carrés.

- En répétant toute la démarche en 5 étapes.
- Avec la méthode **Fit[.]** de *Mathematica*.

§ 3.4 Ajustements trigonométriques

L'ajustement de données périodiques fait généralement appel aux fonctions trigonométriques.

■ Exemple

Ajustez les paramètres c_0, \dots, c_5 de la fonction

$$f(x) = c_0 + c_1 \cos(x) + c_2 \sin(x) + c_3 \cos(2x) + c_4 \sin(2x) + c_5 \cos(3x)$$

aux données

$$\left\{ (0, 0), \left(\frac{2\pi}{9}, 0.2 \right), \left(2\frac{2\pi}{9}, 0.5 \right), \left(3\frac{2\pi}{9}, 0.3 \right), \left(4\frac{2\pi}{9}, -0.2 \right), \right. \\ \left. \left(5\frac{2\pi}{9}, -0.4 \right), \left(6\frac{2\pi}{9}, -0.1 \right), \left(7\frac{2\pi}{9}, 0.1 \right), \left(8\frac{2\pi}{9}, -0.1 \right) \right\}$$

Dans un premier temps, résolvons ce problème comme un ajustement au sens des moindres carrés selon la méthode générale du § 3.1:

$$\mathbf{p} = \left\{ \{0, 0\}, \left\{ \frac{2\pi}{9}, 0.2 \right\}, \left\{ 2\frac{2\pi}{9}, 0.5 \right\}, \left\{ 3\frac{2\pi}{9}, 0.3 \right\}, \right. \\ \left. \left\{ 4\frac{2\pi}{9}, -0.2 \right\}, \left\{ 5\frac{2\pi}{9}, -0.4 \right\}, \left\{ 6\frac{2\pi}{9}, -0.1 \right\}, \left\{ 7\frac{2\pi}{9}, 0.1 \right\}, \left\{ 8\frac{2\pi}{9}, -0.1 \right\} \right\};$$

$$\mathbf{x} = \text{Transpose}[\mathbf{p}][[1]]$$

$$\left\{ 0, \frac{2\pi}{9}, \frac{4\pi}{9}, \frac{2\pi}{3}, \frac{8\pi}{9}, \frac{10\pi}{9}, \frac{4\pi}{3}, \frac{14\pi}{9}, \frac{16\pi}{9} \right\}$$

$$\mathbf{n} = \text{Length}[\mathbf{x}]$$

$$9$$

Les vecteurs de base de l'espace sur lequel on effectue la projection sont

$$\mathbf{g}[0] = \text{Table}[1, \{\mathbf{n}\}]$$

$$\{1, 1, 1, 1, 1, 1, 1, 1, 1\}$$

$$\mathbf{g}[1] = \text{Cos}[\mathbf{x}]$$

$$\left\{ 1, \cos\left[\frac{2\pi}{9}\right], \sin\left[\frac{\pi}{18}\right], -\frac{1}{2}, -\cos\left[\frac{\pi}{9}\right], -\cos\left[\frac{\pi}{9}\right], -\frac{1}{2}, \sin\left[\frac{\pi}{18}\right], \cos\left[\frac{2\pi}{9}\right] \right\}$$

$$\mathbf{g}[2] = \text{Sin}[\mathbf{x}]$$

$$\left\{ 0, \sin\left[\frac{2\pi}{9}\right], \cos\left[\frac{\pi}{18}\right], \frac{\sqrt{3}}{2}, \sin\left[\frac{\pi}{9}\right], -\sin\left[\frac{\pi}{9}\right], -\frac{\sqrt{3}}{2}, -\cos\left[\frac{\pi}{18}\right], -\sin\left[\frac{2\pi}{9}\right] \right\}$$

$$\mathbf{g}[3] = \text{Cos}[2\mathbf{x}]$$

$$\left\{ 1, \sin\left[\frac{\pi}{18}\right], -\cos\left[\frac{\pi}{9}\right], -\frac{1}{2}, \cos\left[\frac{2\pi}{9}\right], \cos\left[\frac{2\pi}{9}\right], -\frac{1}{2}, -\cos\left[\frac{\pi}{9}\right], \sin\left[\frac{\pi}{18}\right] \right\}$$

$$\mathbf{g}[4] = \text{Sin}[2\mathbf{x}]$$

$$\left\{ 0, \cos\left[\frac{\pi}{18}\right], \sin\left[\frac{\pi}{9}\right], -\frac{\sqrt{3}}{2}, -\sin\left[\frac{2\pi}{9}\right], \sin\left[\frac{2\pi}{9}\right], \frac{\sqrt{3}}{2}, -\sin\left[\frac{\pi}{9}\right], -\cos\left[\frac{\pi}{18}\right] \right\}$$

$$\mathbf{g}[5] = \text{Cos}[3\mathbf{x}]$$

$$\left\{1, -\frac{1}{2}, -\frac{1}{2}, 1, -\frac{1}{2}, -\frac{1}{2}, 1, -\frac{1}{2}, -\frac{1}{2}\right\}$$

$$k = 6$$

$$6$$

Le système des équations normales est

```
m = Table[FullSimplify[g[i].g[j]], {j, 0, k - 1}, {i, 0, k - 1}]; MatrixForm[m]
```

$$\begin{pmatrix} 9 & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{9}{2} & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{9}{2} & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{9}{2} & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{9}{2} & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{9}{2} \end{pmatrix}$$

```
y = Transpose[p][[2]]
```

```
{0, 0.2, 0.5, 0.3, -0.2, -0.4, -0.1, 0.1, -0.1}
```

```
v = Table[g[j].y, {j, 0, k - 1}]
```

```
{0.3, 0.644609, 1.00157, -1.10608, -0.0427173, 0.15}
```

```
c = LinearSolve[m, v]
```


```
{0.0333333, 0.143246, 0.222572, -0.245795, -0.00949273, 0.0333333}
```

La fonction d'ajustement est la meilleure approximation des données par un polynôme trigonométrique

```
f[t_] = c.{1, Cos[t], Sin[t], Cos[2 t], Sin[2 t], Cos[3 t]}
```

```
0.0333333 + 0.143246 Cos[t] - 0.245795 Cos[2 t] +  
0.0333333 Cos[3 t] + 0.222572 Sin[t] - 0.00949273 Sin[2 t]
```

```
Plot[f[t], {t, 0, 2 π}, Epilog -> {PointSize[0.02], Point /@ p},  
PlotRange -> 1.2 {Min[y], Max[y]}]
```



■ Première idée de la méthode "transformée de Fourier rapide"

Dans le cas où il y a k fonctions de base

$$1, \cos(x), \sin(x), \cos(2x), \sin(2x), \cos(3x), \sin(3x), \cos(4x), \dots$$

et n abscisses uniformément réparties sur une période

$$0, \frac{2\pi}{n}, 2\frac{2\pi}{n}, 3\frac{2\pi}{n}, \dots, (n-2)\frac{2\pi}{n}, (n-1)\frac{2\pi}{n}$$

avec $k \leq n$, alors la méthode d'ajustement donne une matrice diagonale. La résolution du système nécessite alors moins de calculs. En poursuivant l'effort d'optimisation, il est possible d'organiser les calculs de telle manière que, même appliquée à des données volumineuses, la méthode reste efficace. On parle alors de "*transformée de Fourier rapide*".

■ Généralisation

Fonctions de base de Fourier

Généralisons à des fonctions périodiques de période T . En posant $\omega = \frac{2\pi}{T}$, choisissons les k fonctions suivantes

$$\begin{aligned} f_0(t) &= 1, \\ f_1(t) &= \cos(\omega t), \quad f_2(t) = \sin(\omega t), \\ f_3(t) &= \cos(2\omega t), \quad f_4(t) = \sin(2\omega t), \\ f_5(t) &= \cos(3\omega t), \quad f_6(t) = \sin(3\omega t), \\ f_7(t) &= \cos(4\omega t), \quad \dots, \\ f_{k-1}(t) &= \dots \end{aligned}$$

c'est-à-dire

$$f_0(t) = 1, \quad f_j(t) = \begin{cases} \cos\left(\left(\frac{j+1}{2}\right)\omega t\right) & \text{si } j \text{ est impair} \\ \sin\left(\left(\frac{j}{2}\right)\omega t\right) & \text{si } j \text{ est pair} \end{cases}, \quad j = 1, 2, \dots, k-1$$

Echantillonnage ou discrétisation (ou "digitalisation")

Nous supposons que les abscisses d'échantillonnage sont uniformément réparties sur une période. Plus précisément, dans l'intervalle de référence $[0, T]$, on choisit

$$t_j = j \frac{T}{n}, \quad j = 0, 1, \dots, n-1$$

On suppose que $k \leq n$.

En ces abscisses, on donne les valeurs correspondantes

$$Y_0, Y_1, Y_2, \dots, Y_{n-1}$$

Vecteurs de base

Les vecteurs de base de l'espace vectoriel sur lequel on effectue la projection orthogonale sont $\{\vec{g}_0, \vec{g}_1, \dots, \vec{g}_{k-1}\}$

$$\vec{g}_0 = \begin{pmatrix} 1 \\ 1 \\ 1 \\ 1 \\ \dots \\ 1 \end{pmatrix}, \quad \vec{g}_1 = \begin{pmatrix} \cos(0) \\ \cos\left(\frac{2\pi}{n}\right) \\ \cos\left(2\frac{2\pi}{n}\right) \\ \cos\left(3\frac{2\pi}{n}\right) \\ \dots \\ \cos\left(\frac{(n-1)2\pi}{n}\right) \end{pmatrix}, \quad \dots, \quad \vec{g}_3 = \begin{pmatrix} \cos(0) \\ \cos\left(\frac{4\pi}{n}\right) \\ \cos\left(2\frac{4\pi}{n}\right) \\ \cos\left(3\frac{4\pi}{n}\right) \\ \dots \\ \cos\left(\frac{(n-1)4\pi}{n}\right) \end{pmatrix}, \quad \dots, \quad \vec{g}_{k-1} = \begin{pmatrix} f_{k-1}(t_0) \\ f_{k-1}(t_1) \\ f_{k-1}(t_2) \\ f_{k-1}(t_3) \\ \dots \\ f_{k-1}(t_{n-1}) \end{pmatrix}$$

Théorème 3.4

Avec les choix effectués ci-dessus, les vecteurs $\{\vec{g}_0, \vec{g}_1, \dots, \vec{g}_{k-1}\}$ sont orthogonaux, c'est-à-dire

$$\vec{g}_i \cdot \vec{g}_j = 0 \quad \text{pour} \quad i \neq j$$

En d'autres termes, les équations normales sont un système diagonal. (Voir exercices).

Corollaire

Pour ajuster la fonction

$$f(t) = c_0 f_0(t) + c_1 f_1(t) + \dots + c_{k-1} f_{k-1}(t)$$

aux données, le calcul des coefficients peut s'effectuer simplement au moyen de la formule

$$c_j = \frac{\vec{y} \cdot \vec{g}_j}{\vec{g}_j \cdot \vec{g}_j}, \quad j = 0, 1, \dots, k-1$$

Ces coefficients sont appelés "coefficients de Fourier".

La fonction d'ajustement est périodique de période T .

Les résidus suivants sont minimaux au sens des moindres carrés:

$$e_0 = y_0 - f(t_0), \quad \dots, \quad e_{k-1} = y_{k-1} - f(t_{k-1})$$

Application

Dans le cas où le nombre de points d'échantillonnage est élevé, la résolution des équations normales peut prendre un temps excessif. On peut éviter de former la matrice **m**, le vecteur **v** et d'exécuter **LinearSolve[m, v]**. En lieu et place, on utilise la forme explicite de la réponse:

$$c = \text{Table}\left[\frac{y \cdot g[j]}{g[j] \cdot g[j]}, \{j, 0, k-1\}\right]$$

$$\{0.0333333, 0.143246, 0.222572, -0.245795, -0.00949273, 0.0333333\}$$

Il est encore possible d'améliorer la vitesse de calcul (par exemple, on peut donner les valeurs de $g[i] \cdot g[j]$) mais cela sortirait du cadre de ce cours. La méthode de calcul optimisée s'appelle "transformée de Fourier rapide" (en anglais "Fast Fourier Transform" abrégé FFT).

Cas particulier de l'interpolation

Dans le cas où $k = n$, les résidus sont nuls et la fonction d'ajustement passe par les points donnés:

$$y_0 = f(t_0), \quad \dots, \quad y_{n-1} = f(t_{n-1})$$

Au lieu d'ajustement, on parle alors d'interpolation. Un chapitre entier est consacré à ce thème.

■ Exercice 3.4 - 1

Voici le tableau des températures diurnes moyennes collectées en un lieu imaginaire

Jan	Févr	Mars	Avr	Mai	Juin	Juil	Août	Sept	Oct	Nov	Dec
-1	-2	3	5	8	12	15	14	10	8	5	0

A ces données, ajustez les paramètres de la fonction

$$f(t) = c_0 + c_1 \cos(\omega t) + c_2 \sin(\omega t)$$

Utilisez la méthode de calcul "rapide".

Dans un même graphique, superposez les données et la fonction ajustée.

Indications:

Pour le temps, on prendra le mois comme unité. Nous aurons donc

$$T = 12; \quad \omega = \frac{2\pi}{12}$$

Pour respecter les abscisses imposées par la méthode, nous prendrons $t = 0$ à la mi-janvier;...; $t = 11$ à la mi-décembre.

■ Exercice 3.4 - 2

Ecrivez explicitement les assertions du théorème 3.4 sous la forme d'énoncés trigonométriques. Par exemple,

$$\vec{g}_0 \cdot \vec{g}_1 = \sum_{i=0}^{n-1} \cos \left(i \cdot \frac{2\pi}{n} \right) = 0$$

■ Exercice 3.4 - 3 [Facultatif]

Démontrez les relations suivantes

$$\vec{g}_0 \cdot \vec{g}_1 = \sum_{i=0}^{n-1} \cos \left(i \cdot 1 \cdot \frac{2\pi}{n} \right) = 0$$

$$\vec{g}_0 \cdot \vec{g}_2 = \sum_{i=0}^{n-1} \sin \left(i \cdot 1 \cdot \frac{2\pi}{n} \right) = 0$$

$$\vec{g}_0 \cdot \vec{g}_3 = \sum_{i=0}^{n-1} \cos \left(i \cdot 2 \cdot \frac{2\pi}{n} \right) = 0$$

$$\vec{g}_0 \cdot \vec{g}_4 = \sum_{i=0}^{n-1} \sin \left(i \cdot 2 \cdot \frac{2\pi}{n} \right) = 0$$

$$\vec{g}_0 \cdot \vec{g}_5 = \sum_{i=0}^{n-1} \cos \left(i \cdot 3 \cdot \frac{2\pi}{n} \right) = 0, \quad \text{etc.}$$

Indications

Le point $O(0, 0)$ désignant l'origine, définissons les points E_i^j tels que

$$\vec{OE}_i^j = \begin{pmatrix} \cos \left(i \cdot j \cdot \frac{2\pi}{n} \right) \\ \sin \left(i \cdot j \cdot \frac{2\pi}{n} \right) \end{pmatrix}$$

En prenant les exemples $j = 1$, $j = 2$, $j = 3$, représentez graphiquement et vérifiez géométriquement que

$$\sum_{i=0}^{n-1} \vec{OE}_i^j = \sum_{j=0}^{n-1} \begin{pmatrix} \cos \left(i \cdot j \cdot \frac{2\pi}{n} \right) \\ \sin \left(i \cdot j \cdot \frac{2\pi}{n} \right) \end{pmatrix} = \vec{0}$$

Déduisez que, pour chaque j ,

$$\sum_{i=0}^{n-1} \cos \left(i \cdot j \cdot \frac{2\pi}{n} \right) = 0 \quad \text{et} \quad \sum_{i=0}^{n-1} \sin \left(i \cdot j \cdot \frac{2\pi}{n} \right) = 0$$

■ Exercice 3.4 - 4 [Facultatif]

Démontrez que les vecteurs du théorème 3.4

$$\vec{g}_0, \vec{g}_1, \dots, \vec{g}_{k-1}$$

sont orthogonaux.

Indications

Utilisez les résultats de l'exercice 3.4-3 et les relations trigonométriques

$$\cos(x) \cos(y) = \frac{1}{2} (\cos(x+y) + \cos(x-y))$$

$$\cos(x) \sin(y) = \frac{1}{2} (\sin(x+y) - \sin(x-y))$$

$$\sin(x) \sin(y) = \frac{1}{2} (-\cos(x+y) + \cos(x-y))$$

■ Exercice 3.4 - 5 [Travail dirigé]

Dans la région de Zurich, la température moyenne mensuelle en degrés Celsius est la suivante:

-1	1	4	8	12	15	18	17	14	9	4	1
Janvier	Février	Mars	Avril	Mai	Juin	Juillet	Août	Septembre	Octobre	Novembre	Décembre

Question a)

Ajustez la famille de fonctions

$$f(t) = a + b \cos(\omega t) + c \sin(\omega t)$$

à ces températures.

Indications pour la question a)

- 1) Pour ajuster les 12 points donnés par une simple sinusoïde, nous ne prenons que les 3 premières fonctions de base données dans la théorie : 1 , $\cos(\frac{2\pi}{365}t)$, $\sin(\frac{2\pi}{365}t)$.
- 2) Associer les températures au 15 de chaque mois ne convient qu'approximativement car les mois ont des durées inégales; par ailleurs, l'ajustement se calcule plus aisément pour la suite des abscisses $0, \frac{T}{12}, \frac{2T}{12}, \dots, \frac{11T}{12}$; divisons l'année de 365 jours en 12 intervalles égaux

$$\mathbf{x} = \text{Range}\left[0, 360, \frac{365}{12}\right]$$

$$\left\{0, \frac{365}{12}, \frac{365}{6}, \frac{365}{4}, \frac{365}{3}, \frac{1825}{12}, \frac{365}{2}, \frac{2555}{12}, \frac{730}{3}, \frac{1095}{4}, \frac{1825}{6}, \frac{4015}{12}\right\}$$

ce qui nous donne les points d'interpolation suivants

t [d]	0	$\frac{365}{12}$	$\frac{365}{6}$	$\frac{365}{4}$	$\frac{365}{3}$	$\frac{1825}{12}$	$\frac{365}{2}$	$\frac{2555}{12}$	$\frac{730}{3}$	$\frac{1095}{4}$	$\frac{1825}{6}$	$\frac{4015}{12}$
y [°C]	-1	1	4	8	12	15	18	17	14	9	4	1

- 3) pour interpréter les résultats du calcul, on se réfère au tableau suivant;
il s'agit d'associer les dates voulues à la numérotation des jours par la variable t (ou x)

$$\mathbf{xr} = \text{Round}[\mathbf{x}]$$

$$\{0, 30, 61, 91, 122, 152, 182, 213, 243, 274, 304, 335\}$$

Question b)

Ajustez la fonction

$$f(t) = a + d \cos(\omega t - \varphi)$$

à ces températures.

Indications pour la question b)

Le point de départ est le résultat de la partie a)

$f(t) = a + b \cos(\omega t) + c \sin(\omega t)$ que nous allons mettre sous la forme désirée. Pour ce faire, calculons les coordonnées polaires du vecteur suivant

$$\begin{pmatrix} b \\ c \end{pmatrix} = d \begin{pmatrix} \cos(\varphi) \\ \sin(\varphi) \end{pmatrix}$$

où d est la norme du vecteur (d est dénommé *amplitude*)

$$d = \sqrt{b^2 + c^2}$$

φ désigne l'angle entre les vecteurs $\begin{pmatrix} 1 \\ 0 \end{pmatrix}$ et $\begin{pmatrix} b \\ c \end{pmatrix}$:

$$\cos(\varphi) = \frac{b}{d}, \quad \sin(\varphi) = \frac{c}{d}$$

L'angle φ , dénommé *déphasage*, peut s'écrire

$$\varphi = \text{sign}(c) \cdot \arccos\left(\frac{b}{d}\right)$$

En utilisant la relation trigonométrique (voir *Formulaires et tables*)

$$\cos(\alpha - \beta) = \cos(\alpha) \cos(\beta) + \sin(\alpha) \sin(\beta)$$

la fonction ajustée peut être transformée sous la forme souhaitée

$$\begin{aligned} f(t) &= a + d \left(\cos(\omega t) \frac{b}{d} + \sin(\omega t) \frac{c}{d} \right) = \\ &= a + d (\cos(\omega t) \cos(\varphi) + \sin(\omega t) \sin(\varphi)) = a + d \cos(\omega t - \varphi) \end{aligned}$$

Question c)

Donnez une interprétation physique et graphique des grandeurs suivantes

- 1) a = température annuelle moyenne (en degrés Celsius);
- 2) d = amplitude thermique annuelle (en degrés Celsius);
- 3) $\frac{\varphi}{\omega}$ = décalage temporel (en jours);
- 4) φ = déphasage (en radians).